#-------------------------------------------------
#
# Project created by QtCreator 2013-09-04T20:19:02
#
#-------------------------------------------------

QT       += core gui

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

TARGET = MapExtractor
TEMPLATE = app


CONFIG(debug, debug|release) {
    BUILD_DIR = $$PWD/../../Build/Tools/Debug
    TARGET = $$join(TARGET,,,_d)
    LIBS += -lQtExt_d -ljpeglib_d
} else {
    BUILD_DIR = $$PWD/../../Build/Tools/Release
    LIBS += -lQtExt -ljpeglib
}
contains(TEMPLATE, "lib") {
    DESTDIR = $$BUILD_DIR/lib
    DLLDESTDIR = $$BUILD_DIR/bin
} else {
    DESTDIR = $$BUILD_DIR/bin
}

LIBS += -L$$BUILD_DIR/lib

UI_DIR = $$BUILD_DIR/build/ui/$$TARGET
OBJECTS_DIR = $$BUILD_DIR/build/obj/$$TARGET
MOC_DIR = $$BUILD_DIR/build/moc/$$TARGET
RCC_DIR = $$BUILD_DIR/build/rcc/$$TARGET


INCLUDEPATH += ./include
INCLUDEPATH += ../QtExt/include
INCLUDEPATH += ../jpeglib/include

SOURCES += \
    source/MapExtractorDialog.cpp \
    source/main.cpp \
    source/MapExtractor.cpp

HEADERS  += \
    include/MapExtractorDialog.h \
    include/MapExtractor.h

FORMS    += \
    ui/MapExtractorDialog.ui
